First publ. in: Behavior Research Methods ; 23 (1991), 1. - S. 72-76

http://dx.doi.org/10.3758/BF03203337

PSYCHOBASIC: A BASIC dialect for the control of psychological experiments with the Commodore-64 and DELA interfacing

L. XIA, C. D. L. WYNNE, F. VON MÜNCHOW-POHL, and J. D. DELIUS
University of Konstanz, Konstanz, Germany

A modified version of BASIC for the control of psychological experiments is presented. PSYCHOBASIC runs on Commodore-64 computers with DELA interfacing, and it is based on Commodore BASIC. New commands control digital and analog input and output, centisecond timers, and counters. A control panel, which is displayed on the screen at all times, gives the status of many PSYCHOBASIC components. The interpreted nature of PSYCHOBASIC, as well as the control panel, makes the system easy to learn and use. The PSYCHOBASIC system is less expensive than equivalent systems for IBM-PC computers. A sample PSYCHOBASIC program is given.

The use of computers for the control of psychological experiments has a history covering nearly three decades (Leslie, 1981). Recently two systems SKED and MEDSTATE, both of which are based on the state-space notation system, have become dominant. The state-space approach is a clear method for describing experimental conditions, but it entails a number of difficulties. We were motivated to develop our own system by two drawbacks in the extant systems. First, they were too expensive; SKED and MEDSTATE are based on the IBM-PC computer system with special interfacing (a minimum investment of over $2900 for 16 I/O lines with software). Second, the two-stage compilation system of the state-space systems causes undesirable delays between the typing of a command and seeing its effect. We have found that many students and even colleagues still prefer to work with the solid-state logic units because they provide immediate feedback about their logical state.

We therefore had three objectives: The system should (1) require a minimum of new learning of programming techniques; (2) be inexpensive enough to make it unnecessary to multitask several experiments on one computer, yet reliable enough to take the strain of dusty animal rooms; and (3) as far as was possible, provide immediate feedback of its state.

Our solution is based on the Commodore-64 (C-64) computer with DELA interfacing and the BASIC programming language. The C-64 is the cheapest computer that offers sufficient reliability for experimental purposes, and it is widely available. The DELA interface system[1] (Wynne, 1990) connects to the user port of a C-64 and controls up to 384 lines of I/O. The combined cost of a C-64 with disk drive and sufficient DELA interfacing for 24 input lines (TTL) and 24 output lines (relays capable of switching 28-V dc or 120-V ac) is at present under $600. The PSYCHOBASIC control language is an adaptation of the Commodore-64 BASIC V2 supplied as standard with the C-64. Although BASIC is neither the most elegant nor the fastest programming language available, it has the advantage of being widely known. The factor that is surely responsible for the success of BASIC as a programming language for beginners has been carried over in PSYCHOBASIC; the language is interpreted instead of compiled. This means that commands can be typed into the computer and executed immediately. This is especially useful for the on-line control of experiments, because it makes it possible for the experimenter to test hardware without the need to write and compile test programs. PSYCHOBASIC also incorporates the most useful feature of solid-state programming systems: immediate visual feedback in response to most I/O commands.

PSYCHOBASIC OPERATION

PSYCHOBASIC consists of two programs: a BASIC loader, PSYCHO.BAS, and PSYCHOBASIC itself, PSYCHO.EXE. PSYCHO.BAS is loaded from disk as a normal BASIC program; running this program then causes PSYCHO.EXE to be loaded into memory. This involves copying the Commodore BASIC from ROM into RAM and adding in the PSYCHOBASIC commands. It takes PSYCHO.BAS up to 3 min to load PSYCHOBASIC and display the PSYCHOBASIC control panel at the bottom of the screen. To identify itself, PSYCHOBASIC also replaces the standard BASIC "Ready." message with an "O.K." response after each command. PSYCHOBASIC

Equipment was made available through grants from the Deutsche Forschungsgemeinschaft to J. D. Delius and from the National Science Foundation to John Staddon at Duke University. C. D. L. Wynne was supported by the Alexander von Humboldt-Stiftung. Address correspondence either to J. D. Delius at Universität Konstanz, Allgemeine Psychologie, Postfach 5560, D-7750 Konstanz, Germany, or to C. D. L. Wynne, Department of Psychology, Duke University, Durham, NC 27706.

occupies the free RAM area from 49152 ($C000) to 53247 ($CFFF).

All Commodore BASIC commands work within PSYCHOBASIC in the normal way. Ten additional new commands control input, output, counters, and timers. These commands are controlled by a new 100-Hz interrupt. Every 10 msec, BASIC operation is interrupted to monitor the status of the PSYCHOBASIC timers, counters, and inputs; this takes about 2 msec (maximum 2.8 msec). The duration of the interrupt depends on the number of timers, counters, and inputs in operation, and on whether or not an event is detected. If the PSYCHOBASIC interrupt detects an event that requires servicing, it stores this information and then services the event before interpreting the next line of BASIC. The input pulses from the DELA interfaces are long enough so that inputs cannot be lost between interrupt cycles. PSYCHOBASIC senses whether the computer in use is a North American or European model of the C-64 and adjusts its calculations to take into account the differing clock frequencies, so the user does not need to be concerned with such technicalities.

PSYCHOBASIC SYSTEM COMPONENTS

PSYCHOBASIC Control Panel

During PSYCHOBASIC operation, the bottom six lines of the video screen are occupied by the PSYCHOBASIC Control Panel (Figure 1). This shows the status of the first 24 digital input channels, the first 24 digital output channels, the first two analog/digital (A/D) converters, the first two digital/analog (D/A) converters, the first five timers, and the first five counters. The A/D, D/A, counter, and timer status are presented digitally (in volts for the A/D and D/A converters, in events counted for the counters, and in seconds for the timers). The input and output sections of the control panel each contain 24 digits, each representing one of the input or output channels. In the passive state (output channel off, input line open), the relevant digit appears in normal video (white on a black background). When the output channel is switched on, or the input line is closed, the corresponding digit on the control panel appears in reversed video (black on a white background). Through the control panel, a program may be tested without the need for the computer to be connected to an interface board; similarly, the input section of the control panel makes it possible to test a manipulandum without the need for any programming.



Figure 1. PSYCHOBASIC control panel.

The DELA Interfaces

The drawback of the C-64 as a computer for on-line experimentation has been the limited number of lines on the user port (10 I/O lines, O'Dell & Jackson, 1986). The great advantage of the DELA interface system is that it utilizes 2 of the lines on the user port to control the other 8, thus substantially increasing the number of lines (see Wynne, 1990, for more details). The most interesting interface cards for experimental control are the Relay output cards containing eight relays, and the TTL input cards with 24 input lines each. PSYCHOBASIC also controls A/D and D/A converter cards which each cover 5 V in 256 0.02-V steps, and TTL output cards, each with 24 input lines. Each DELA interface card contains a DIP switch, by means of which the card address can be set. PSYCHOBASIC allocates logical channel numbers to up to 16 DELA cards, which can be connected together and to the user port by means of IC sockets on the interface cards.

Hardware

At present, PSYCHOBASIC will run on any model of the C-64 computer with monitor and disk drive (Commodore 1541, 1541-II, 1571, or compatible clone model) and DELA interfacing. The system is not compatible with the Commodore 128 (even in C-64 mode) or with other interfaces. An external 5-V dc supply will also be required, and additional interfacing will be necessary if the inputs are not at TTL levels (O'Dell & Jackson, 1986; Wynne, 1990).

SOFTWARE

PSYCHOBASIC Commands and Functions

In the following description of the PSYCHOBASIC commands, the parameters in *italics* must be replaced with values; text in **bold** type must be entered exactly as shown; square brackets show optional parameters. All commands operate in immediate as well as program mode. The commands TMO, TRG, and OVF all implicitly use the BASIC GOSUB construction, so that the subroutines that they call must terminate with a RETURN command. The programmer must also consider what should happen if an additional event is found by TMO, TRG, or OVF while the subroutine is processing an external event.

OUTPUT: Digital

OUT *n*, **1**
Switch output channel *n* on

OUT *n*, **0**
Switch output channel *n* off

OUTPUT: Analog

DAC *n*, *v*
Sets the desired voltage *v* ($0 < v < 5$) on the selected D/A converter *n*.

DAC(*n*)
Returns the present state of D/A converter *n* in volts.

INPUT: Digital

INP(*n*)
Returns the status of input line *n*. INP(*n*) is 0 if the input line is open (low) and 1 if the input line is closed (high).

TRG[±]*n*, *Line-number*
An event on input line *n* moves program control to line number *Line-number*. Optional + or − determines whether input event should be positive slope (closure of input line) or negative slope. This command includes a debounce of approximately 40 msec.

TRG*n*, 0
Removes previous TRG command on input line *n*.

TRG([±]*n*)
This command returns a 1 if an event has occurred on input line *n*. Entering + or − specifies whether the event should have a positive or a negative slope. TRG(0) returns the number of the input line on which the last event took place.

The command TRG is interrupt-driven. Issued once, it remains in force until another TRG command is issued for the same input line. Care must be taken to switch off TRG commands (with **TRG*n*, 0**) when they are no longer needed and to "keep the program occupied"—for example, with an empty loop—if no other operations are required except waiting for an input. TRG operates only with the input channels 1-24. The subroutine to which TRG jumps when an event occurs must terminate with RETURN; control then passes to the next command after the call to TRG.

INPUT: Analog

ADC(*n*)
Returns the voltage on A/D card *n*.

TIMER

TMO*n*, *interval*, *Linenumber*
After *interval* seconds has elapsed on timer *n*, GOSUB to line number *Linenumber*.

TMO*n*, *interval*, 0
Start timer *n* to count down *interval* seconds.

Use this form of the TMO command, in combination with the TMR command to measure reaction times.

TMR(*n*)
Returns present state of timer *n*.

Eight centisecond timers are available. The maximum interval duration is 167,000 sec.

COUNTER

PSYCHOBASIC contains eight counters that are controlled with the following three commands.

CNT*n*, [±]*m*[,*m2*,*m3*,....]
Sets counter *n* to count events on input line[s] *m* [*m2*,*m3*,...]. Inserting + or − makes it possible to determine whether inputs with positive or negative slope should be counted. Only the first 24 inputs may be used with this command.

CNT*n*, 0
Stops counting of input events on counter *n*; the present stand of counter *n* is unaffected.

CTR*n*, *NewValue*
Sets counter *n* to the value *NewValue*. The counter can take positive values up to 65535.

CTR(*n*)
Returns state of counter *n* so long as the corresponding CNT command has been given.

OVF *n*, *TerminalValue*, *LineNumber*
Determines the action when an overflow occurs. When counter *n* reaches the value *TerminalValue*, control passes to line number *LineNumber*. The subroutine in *LineNumber* must terminate with RETURN. OVF clears itself after operation.

OVF *n*, 0
Deletes a previous Overflow definition.

MISCELLANEOUS

<CTRL> <RESTORE>
Switch PSYCHOBASIC out, return to Commodore BASIC.

<RUN/STOP> <RESTORE>
Warm start: switch PSYCHOBASIC on (if in Commodore BASIC), clear screen (if already in PSYCHOBASIC).

These two commands do not delete a BASIC or PSYCHOBASIC program from memory.

RID.PSY: A Sample PSYCHOBASIC Program

The appended listing contains a sample program, RID.PSY, which demonstrates some of the capabilities of PSYCHOBASIC. The program controls a response-initiated delay (RID) schedule in a Skinner box. An RID schedule (Wynne & Staddon, 1988) is equivalent to a chain FR1 FT: The first response within each trial starts a timer that times out with presentation of food at the end of a fixed time interval, independently of further responding. RID.PSY is set up for a pigeon box with a single response key (with red and green keylights), a houselight, and a food hopper. Lines 100-130 switch the houselight on and the keylights off through use of OUT statements. Line 250 starts timer 1 to count down from 100,000 sec. This timer is used to time event times throughout the session. Line 270 sends program control to line 400, where the first phase of each trial starts. The empty loop in line 280 is extremely important: it occupies the program while it is waiting for an input or other event. The first phase of each trial is programmed in lines 400 ff. Here the pecking key is illuminated red (OUT statement in line 420); line 410 records the time of the start of the trial in the array R. The TRG command in line 430 assures that as soon as a peck occurs control is sent to line 500. The RETURN command in line 440 sends control back to line 280 where the program waits in an empty loop until an input on line 1 activates the TRG command in line 430.

Once a peck occurs, control goes to the subroutine in lines 500 ff. The red keylight is extinguished and replaced by the green keylight (OUT commands in lines 550 and 570, respectively). The critically important action of removing the TRG command that brought program control to line 500 is carried out by the TRG1,0 command in line 510. Line 520 records the time of the change from red to green key illumination in the array R. The TMO command in line 580 ensures that when the requisite delay has expired, control passes to line 600. Line 590 sets up a TRG command that causes command to pass to lines 800 ff whenever a peck occurs (these lines simply record the time of the event in array R and return command to line 590). After time DE has expired, command passes to the subroutine in lines 600 ff. The TRG command sending keypecks to lines 800 ff is cleared (in line 610). Then the green keylight is turned off, and the feeder is put in the up position through OUT commands in lines 610 and 640. This subroutine ends with another TMO command, this time sending control, after RF sec, to line 700. At lines 700 ff, a small subroutine puts the feeder in the down position (line 710), decrements the trial counter (line 730), and sends command either back to the start of the main loop or on to the subroutine at lines 1000 ff, depending on whether or not any trials remain to be run. The last section of the program (lines 1000 ff) turns out the houselight (line 1010) and then prints the data collected onto the screen. The times at the start of each trial and of the first peck within each trial are coded as negative; pecks within the green phase appear as positive. Obviously these data could be printed out on a printer, recorded on a floppy disk, or sent through a serial interface to another computer for later analysis.

AVAILABILITY

A disk containing PSYCHO.BAS and PSYCHO.EXE can be had in Europe for 10 DM (to cover costs) from J. D. Delius; or outside Europe for $10 from C. D. L. Wynne.

REFERENCES

LESLIE, J. C. (1981). State notation programming languages in psychology. *International Journal of Man-Machine Studies*, **14**, 341-354.

O'DELL, J. W., & JACKSON, D. E. (1986). The Commodore 64 and an interface system for controlling operant chambers. *Behavior Research Methods, Instruments, & Computers*, **18**, 339-341.

WYNNE, C. D. L. (1990). A Commodore 64-based interface system for the operant laboratory. *Behavior Research Methods, Instruments, & Computers*, **22**, 27-33.

WYNNE C. D. L., & STADDON, J. E. R. (1988). Typical delay determines waiting time on periodic-food schedules: Static and dynamic tests. *Journal of the Experimental Analysis of Behavior*, **50**, 197-210.

NOTE

1. Available from McByte GmbH, Maastrichter-Str., D-5000 Cologne 1, Germany: Digital Input Card (Model 1005003), Digital Output Card (Model 1005004), Relay Output Card (Model 1005002), D/A converter (Model 1005008), A/D converter (Model 1005007).

LISTING

Sample PSYCHOBASIC Program for the Control of a Response-Initiated-Delay Experiment.

```
10 REM ****************************
20 REM ** RESPONSE-INITIATED-   **
30 REM ** DELAY  RID.PSY.       **
40 REM ** CLIVE WYNNE 1990      **
50 REM ** KONSTANZ FRG          **
60 REM ****************************
90 DIM R(3000):REM ARRAY FOR TIMES
100 OUT1,1:    REM HOUSELIGHT ON
110 OUT2,0:    REM RED LIGHT OFF
120 OUT3,0:    REM GREEN LIGHT OFF
130 N%=50:     REM NO OF TRIALS
140 RF=2:      REM LENGTH OF REINF.
145 PRINT "(CLR)(DOWN)(DOWN)RID EXPERIMENT"
150 INPUT "(DOWN)(DOWN)    DATE";DA
160 IF DA >3112 OR DA<101  GOTO 150
170 INPUT"(UP)(RGHT)(RGHT)(RGHT)(RGHT)(RGHT)(RGHT)(RGHT)(RGHT)(RGHT)(RGHT)
(RGHT)(RGHT)(RGHT)(RGHT)(RGHT)(RGHT)(RGHT)(RGHT)     BIRD";B%
180 INPUT"(DOWN)    FOOD DELAY";DE
230 PRINT"(DOWN)PRESS <RETURN> TO START"
240 INPUT B$
250 TMO1,100000,0:REM SET TIMER 1
270 GOSUB 400
280 GOTO 280
400 REM ****    RED PERIOD     ****
405 PRINT"(UP)RED        N=";51 -N%
410 I=I+1:R(I)=-TMR(1)
420 OUT 2,1:  REM RED ON
430 TRG1,500: REM SET KEY PECK
440 RETURN
500 REM **** GREEN PERIOD      ****
520 PRINT"(UP)GREEN     "
```

```
530 TRG1,0:   REM CLEAR KEY PECK
540 I=I+1:R(I)=-TMR(1)
550 OUT 2,0:  REM RED OFF
560 OUT 3,1:  REM GREEN ON
570 TMO2,DE,600:REM TIME DELAY
580 TRG1,800: REM NEW KEY PECK
590 RETURN
600 REM ***  REINFORCEMENT      ***
610 TRG1,0:   REM CLEAR KEY PECK
620 PRINT"(UP)FOOD     "
630 OUT 3,0:  REM KEY LIGHTS OFF
640 OUT 6,1:  REM FEEDER UP
650 TMO2,RF,700:REM TIME FOOD
660 RETURN
700 REM ***  END REINF.       ****
710 OUT6,0 :   REM FEEDER DOWN
720 TMO2,0:    REM CLEAR TIMER 2
730 N%=N%-1:   REM TRIAL COUNTER
740 IF N%=0 GOTO 1000
750 GOTO 400
800 REM ****  RESPONSE         ****
810 I=I+1
820 R(I)=TMR(1):REM TIME OF PECK
830 RETURN
1000 OUT1,0:    REM HOUSELIGHT OUT
1010 INPUT"WHEN READY PRESS <RETURN>";B$
1020 REM ** PRINT OUT DATA        **
1030 R(0)=I
1040 PRINT "DATE ";DA
1050 PRINT "BIRD ";B%
1060 PRINT "DELAY ";DE
1070 PRINT "EVENT TIME"
1080 FOR I=1 TO R(0):PRINT R(I)
1090 NEXT I
1100 END
```